
Chapter 1

Conventions

For concision, the specifications in this document are typically phrased as im-
plications so that they only apply to expressions that terminate. It is not the
case that all expressions terminate. If a higher order function is applied to ar-
guments including a function and input for that function on which it does not
terminate, then the application will likely not terminate.

The specifications in this document often contain data type definitions and
code fragments in SML-like syntax. These fragments exist to clearly and care-
fully specify the behaviour of the value being documented, not to restrict its
possible implementations. Structures do not need to implement the signature
with exactly these code fragments; they are typically extremely inefficient. They
also contain a mixture of SML syntax and notation from the abstractions in play,
and therefore do not immediately compile.

Two functions f and g are said to be logically equivalent if they give equal
results on equal arguments. It follows that if f is logically equivalent to g, and
f does not terminate on some input, then g does not terminate on that input.

We use math and monospace fonts to indicate the difference between the
abstraction implemented by an abstract type and SML representations of that
abstract notion, respectively. If the same letter or identifier appears in both
fonts, it should be understood to mean either the abstraction or representation
of the same object.

Types are always typeset in a math font for readability, corresponding to
the typical pronunciation of concrete SML syntax. For example, the type of a
function

f : ’a -> (’a * ’b) -> ’c

will be written
f : α → α ∗ β → γ

When talking about a polymorphic expressions in prose, we will often omit
the type variable that they are polymorphic over. For example, we may say
that “l is a list” to mean “l has type α list”.

The specifications often use the phrase “x is a t value”, where t is some type.
For example, when we say“i is an int value” we mean that i is an expression
with type int that cannot be evaluated further, like 7 but not like ((fn x =>

x) 7).
Combining combing the above two conventions, the statement

“s is a sequence value”

should be taken to mean

“s has type α seq, and s is a value”

or, more specifically,

“s has type α seq, and for every valid index i into s, si is a value”

Chapter 2

SEQUENCE Signature

2.1 Overview

2.1.1 Abstract Sequences

We define an abstract mathematical notion of a sequence. The documentation
for the signature that follows states the behaviour of implementations in terms
of this abstraction.

• A sequence is an ordered finite list of elements of some type, indexed by
the natural numbers.

• The length of a sequence is the number of elements in that sequence. If s
is a sequence, its length is denoted |s|.

• A natural number i is said to be a valid index into the sequence s if and
only if 0 ≤ i < |s|.

• If s is any seqence and i is a valid index into s, then si denotes the ith

element of s.

• If s is a particular sequence with n elements, we may denote s with the
notation

〈s0, s1, . . . , sn−1〉

For example,
〈〉

denotes the empty sequence and

〈4, 2, 3〉

denotes a particular sequence of natural numbers with length three.

s0 s1 s2 s3 s4 s5 s6

r0 r1 r2

r3 r4

r5

Figure 2.1: Reduce tree structure for any sequence s with |s| = 7

• A sequence s′ is said to be a subsequence of a sequence s if there is a
strictly increasing, possibly empty, sequence I of valid indices into s such
that s′i = sIi

.

• Sequences can only be ordered if their elements can be ordered. If that
condition is met, sequences are ordered lexicographically.

2.2 Signature Definition

signature SEQUENCE =

sig

type ’a seq

datatype ’a treeview = EMPTY

| ELT of ’a

| NODE of (’a seq * ’a seq)

datatype ’a listview = NIL

| CONS of (’a * ’a seq)

type ’a ord = ’a * ’a -> order

exception Range

val empty : unit -> ’a seq

val singleton : ’a -> ’a seq

val length : ’a seq -> int

val nth : ’a seq -> int -> ’a

val tabulate : (int -> ’a) -> int -> ’a seq

val fromList : ’a list -> ’a seq

val collate : ’a ord -> ’a seq ord

val map : (’a -> ’b) -> ’a seq -> ’b seq

val map2 : ((’a * ’b) -> ’c) -> ’a seq -> ’b seq -> ’c seq

val reduce : ((’a * ’a) -> ’a) -> ’a -> ’a seq -> ’a

val scan : ((’a * ’a) -> ’a) -> ’a -> ’a seq -> (’a seq * ’a)

val filter : (’a -> bool) -> ’a seq -> ’a seq

val iter : (’b * ’a -> ’b) -> ’b -> ’a seq -> ’b

val iterh : (’b * ’a -> ’b) -> ’b -> ’a seq -> (’b seq * ’b)

val flatten : ’a seq seq -> ’a seq

val partition : int seq -> ’a seq -> ’a seq seq

val inject : (int*’a) seq -> ’a seq -> ’a seq

val append : ’a seq * ’a seq -> ’a seq

val take : ’a seq * int -> ’a seq

val drop : ’a seq * int -> ’a seq

val rake : ’a seq -> (int * int * int) -> ’a seq

val subseq : ’a seq -> (int * int) -> ’a seq

val splitMid : ’a seq * int -> (’a seq * ’a * ’a seq) option

val sort : ’a ord -> ’a seq -> ’a seq

val merge : ’a ord -> ’a seq -> ’a seq -> ’a seq

val collect : ’a ord -> (’a * ’b) seq -> (’a * ’b seq) seq

val toString : (’a -> string) -> ’a seq -> string

val tokens : (char -> bool) -> string -> string seq

val fields : (char -> bool) -> string -> string seq

val showt : ’a seq -> ’a treeview

val showti : ’a seq -> (int -> int) -> ’a treeview

val hidet : ’a treeview -> ’a seq

val showl : ’a seq -> ’a listview

val hidel : ’a listview -> ’a seq

val % : ’a list -> ’a seq

end

2.3 Details of Types

2.3.1 α seq

This is the abstract type that represents the notion of a sequence described in
section 2.1.1.

2.3.2 α treeview

α treeview provides a view of the abstract α seq type as a binary tree.

2.3.3 α listview

α listiew provides a view of the abstract α seq type as a list.

2.3.4 α ord

The type α ord represents an ordering on the type α as a function from pairs
of elements of α to order .

2.4 Details of Exceptions

2.4.1 Range

Range is raised whenever an invalid index into a sequence is used. The specifi-
cations for the individual functions state when this will happen more precisely.

This is the only exception that the functions in a module ascribing to
SEQUENCE raise. An expression applying such a function to appropriate ar-
guments may raise other exceptions, but it will do so only because one of the
arguments in that application raised the other exception.

2.5 Details of Values

2.5.1 empty: unit → α seq

(empty ()) evaluates to 〈〉.

2.5.2 singleton: α → α seq

If x is a value, then (singleton x) evaluates to 〈x〉.

2.5.3 length: α seq → int

If s is a sequence value, then (length s) evaluates to |s|.

2.5.4 nth: α seq → int → α

If s is a sequence value and i is an int value and i is a valid index into s, then
(nth s i) evaluates to si.

This application raises Range if i is not a valid index.

2.5.5 tabulate: (int → α) → int → α seq

If f is a function and n is an int value, then (tabulate f n) evaluates to a
sequence s such that |s| = n and, for all valid indicies i into s, si is the result
of evaluating (f i).

Note that the evaluation of this application will only terminate if f termi-
nates on all valid indices into the result sequence s.

2.5.6 fromList: α list → α seq

If l is a list value, then (fromList l) evaluates to the index preserving sequence
representation of l. That is to say, fromList is logically equivalent to

fn l => tabulate (fn i => List.nth(l,i)) (List.length l)

2.5.7 collate: α ord → α seq ord

If ord is an ordering on the type α, collate ord evaluates to an ordering on
the type α seq derived lexicographically from ord.

2.5.8 map: (α → β) → α seq → β seq

If f is a function and s is a sequence value such that |s| = n, then (map f s)

evaluates to the sequence r such that |r| = n and, for all valid indicies i into s,
ri is the result of evaluating (f si).

Note that the evaluation of this application will only terminate if f termi-
nates on si for all valid indicies i.

2.5.9 map2: ((α × β) → γ) → α seq → β seq → γ seq

If f is a function and s1 and s2 are sequence values, then (map2 f s1 s2)

evaluates to the sequence r such that ri is the result of evaluating f (s1i, s2i)
for all i that are valid indices into both s1 and s2.

It follows from the definition of a valid index and the above specification
that

|r| = min(|s1| , |s2|)

Note that the evaluation of this application will only terminate if f termi-
nates on (s1i, s2i) for all 0 ≤ i < |r|.

2.5.10 reduce: ((α × α) → α) → α → α seq → α

To define the behaviour of reduce, we’ll first define a type of non-empty binary
trees, then a mapping from non-empty sequences to those trees, then an analog
to reduce on trees, and finally reduce on sequences.

The type of non-empty trees we’ll use is

datatype ’a tree = Leaf of ’a

| Node of (’a tree * ’a tree)

Assume that prevpow2 is a function with type int → int such that if x is
an int value then prevpow2 x evaluates to the maximum element of the set

{

y|y < x ∧ ∃i ∈ N.y = 2i
}

With these two assumptions, we define a mapping from non-empty sequences
to trees as

fun toTree s =

case |s|
of 1 => Leaf(s0)

| n => Node(toTree (take (s, prevpow2 |s|)),
toTree (drop (s, prevpow2 |s|)))

The result of this is a nearly-balanced tree where the number of leaves to
the left of any internal node is the greatest power-of-two less than the total
number of leaves below that node. The structure of such trees depends only on
the length of the input sequence. An example tree is shown in Figure 2.1.

We’ll now define the function reducet for the tree type. reducet has type

((α × α) → α) → α tree → α

and is defined as

fun reducet f (Leaf x) = x

| reducet f (Node(l,r)) = f(reducet l, reducet r)

Finally, if f is a function, b a value, and s a sequence value, there are two
cases:

• If |s| = 0 then (reduce f b s) evaluates to b.

• If |s| > 0, and (reducet f (toTree s)) evaluates to some value v, then
(reduce f b s) evaluates to f(b, v).

Note that this definition does not require that f is associative. The trans-
formation to trees and reduce on trees are both well-defined without respect to
any associativity of f. The tree structure defined by toTree defines a particular
association of f on any sequence: if we use ⊕ as infix notation for f, the tree
corresponds to exactly one of the many ways to parenthesize the expression

(

s0 ⊕ s1 ⊕ . . . ⊕ s|s|−1

)

⊕ b

If f happens to be associative, all of the possible ways to parenthesize this
expression result in the same computation.

It follows that if f is associative and b is a value such that b is the identity
of f , reduce f b is logically equivalent to iter f b.

2.5.11 scan: ((α × α) → α) → α → α seq → (α seq × α)

If f is an associative function, and b a value such that b is an identity of f ,
(scan f b) is logically equivalent to

fn s =>

(tabulate (fn i => reduce f b (take(s,i))) (length s),

reduce f b s)

2.5.12 filter: (α → bool) → α seq → α seq

If p is a predicate and s is a sequence value, then (filter p s) evaluates to
the longest subsequence s′ of s such that p holds for every element of s′.

2.5.13 iter: (β × α → β) → β → α seq → β

iter is logically equivalent to iterate, defined below.

fun iterate f b s =

case showl s

of NIL => b

| CONS (x,xs) => iter f (f(b,x)) xs

Less formally, if f is a function, b is a value, and s is a sequence value, then
(iter f b s) computes the iteration of f on s with left-association and b as
the base case. We can write this iteration as

f(f(. . . f(b, s0), . . . s|s|−2), s|s|−1)

or, using ⊕ as infix notation for f ,

(. . . (((b ⊕ s0) ⊕ s1) ⊕ s2) ⊕ . . . ⊕ s|s|−1)

2.5.14 iterh: (β × α → β) → β → α seq → (β seq × β)

iterh is a generalization of iter that also computes the sequence of all par-
tial results produced by the iterated application of the functional argument.
Specifically, (iterh f b) is logically equivalent to

fn s => (tabulate (fn i => iter f b (take (i,s))) (|s|),
iter f b s)

2.5.15 flatten: α seq seq → α seq

flatten is logically equivalent to (iter append (empty ())).
Less formally, if s is a sequence value of sequence values, then (flatten

s) evaluates to the concatenation of the sequences in s in the order that they
appear in s.

2.5.16 partition: int seq → α seq → α seq seq

If I is an int sequence value and s is a sequence value, then (partition I

s) evauluates to a sequence of sequences p such that |p| = |I| and, for every i
that’s a valid index of I, pi is a subsequence of s of length Ii starting at index

i−1
∑

j=0

Ij

in s.
That is to say, partition produces a sequence of adjacent subsequence of s

with lengths specified by the elements of I.
Let

l :=
∑

i

Ii

partition has the property that for any sequence s,

(flatten (partition I s))

is the first subsequence of s of length l. In particular, if l = |s|, this means that

fn ss => (partition (map length ss) (flatten ss))

is functionally equivalent to the identity function on tseqα seq.

2.5.17 inject: (int × α) seq → α seq → α seq

Let ind and s be sequence values and let

occ(i) := {j|indj = (i, x) for some x}

(inject ind s) evaluates to the sequence s′ with length |s|, where for all valid
indicies i into s

s′i =

{

si occ(i) = {}
x j = max (occ(i)) ∧ indj = (i, x)

This application will raise Range if any element of ind has a first component
that is not a valid index into s.

2.5.18 append: α seq × α seq → α seq

If s1 and s2 are sequence values, then (append (s1, s2)) evaluates to a se-
quence s with length |s1|+ |s2| such that the subsequence of s starting at index
0 with length |s1| is s1 and the subsequence of s starting at index |s1| with
length |s2| is s2.

2.5.19 take: α seq × int → α seq

If s is a sequence value and n is an integer, then (take (s,n)) evaluates to the
first subsequence of s of length n. This application will raise Range if n > |s|.

2.5.20 drop: α seq × int → α seq

If s is a sequence value and n is an integer, then (drop (s,n)) evaluates to
the last subsequence of s of length |s| − n. This application will raise Range if
n > |s|.

2.5.21 subseq: α seq → (int × int) → α seq

If s is a sequence value and j and len int are values such that j + len < |s|,
then (subseq s (j, len)) evaluates to the subsequence s′ of s of length len
such that s′i = si+j for all i < len.

This application will raise Range if the subsequence specification is invalid.

2.5.22 splitMid: α seq × int → (α seq × α × α seq) option

Let s be a sequence value and i an int value.

• If |s| = 0, then (splitMid(s,i)) evaluates to NONE.

• If |s| > 0, (splitMid(s,i)) evaluates to (SOME (l,si,r)) where l is the
first subsequence of s of length i − 1 and r is the last subsequence of s of
length |s| − 1 − i.

This application will raise Range if i > |s|.

2.5.23 sort: α ord → α seq → α seq

If ord is an ordering and s is a sequence, (sort ord s) evaluates to a rear-
rangement of the elements of s that is sorted with respect to ord.

2.5.24 merge: α ord → α seq → α seq → α seq

2.5.25 collect: α ord → (α × β) seq → (α × β seq) seq

Let ord be an ordering and s be a sequence of pairs. (collect ord s) evaluates
to a sequence of sequences where each unique first coordinate of elements of s is
paired with the sequence of second coordinates of elements of s. The resultant
sequence is sorted by the first coordinates, according to ord. The elements in
the second coordinates appear in their original order in s.

For example, if

s = 〈(5, ”b”), (1, ”a”), (1, ”b”), (1, ”b”)〉

and ord is the usual ordering on integers, then (collect ord s) will evaluate
to

〈(1, 〈”a”, ”b”, ”b”〉), (5, 〈”b”〉)〉

2.5.26 toString: (α → string) → α seq → string

If f is a function and s is a sequence value, (toString f s) evaluates to a string
representation of s. This representation begins with “〈”, which is followed by
the results of applying f to each element of s, in left-to-right order, interleaved
with “,”, and ends with “〉”.

2.5.27 tokens: (char → bool) → string → string seq

Let p be a predicate on characters. A token is a non-empty maximal substring
of a string not containing any character that satisfies p. If p is a predicate on
characters and s is a string, (tokens p s) evaluates to the sequence of tokens
of s in left to right order.

For example, tokens Char.isPunct ‘‘the,,,horse’’ evaluates to

〈“the′′, “horse′′〉

and tokens Char.isPunct ‘‘the,horse’’ evaluates to

〈“the′′, “horse′′〉

.

2.5.28 fields: (char → bool) → string → string seq

Let p be a predicate on characters. A field is a possibly empty maximal substring
of not containing any character that satisfies p. If p is a predicate on characters
and s is a string, (fields p s) evaluates to the sequence of tokens of s in left
to right order.

For example, fields Char.isPunct ‘‘the,,,horse’’ evaluates to

〈“the′′, “′′, “′′, “horse′′〉

and fields Char.isPunct ‘‘the,horse’’ evaluates to

〈“the′′, “horse′′〉

.

2.5.29 showt: α seq → α treeview

Let s be a sequence value.

• If |s| = 0, (showt s) evaluates to EMPTY.

• If |s| = 1, (showt s) evaluates to ELT(s0).

• If |s| > 1, and NODE(take (s, (|s| /2)), drop (s, (|s| /2))) evalautes
to some value v, (showt s) evaluates to v.

2.5.30 showti: α seq → (int → int) → α treeview

Let s be a sequence value.

• If |s| = 0, (showti s f) evaluates to EMPTY.

• If |s| = 1, (showti s f) evaluates to ELT(s0).

• If |s| > 1, f : int → int is a function, and NODE(take (s, f |s|), drop

(s, f |s|)) evalautes to some value v, (showti s f) evaluates to v.

2.5.31 hidet: α treeview → α seq

Let tv be a treeview value.

• If tv is EMPTY, then (hidet tv) evaluates to 〈〉.

• If tv is (ELT x), then (hidet tv) evaluates to 〈x〉.

• If tv is NODE (l,r), then (hidet tv) evaluates to the same value as
append (l,r).

2.5.32 showl: α seq → α listview

Let s be a sequence value.

• If |s| = 0, (showl s) evaluates to NIL.

• If |s| > 0, (showl s) evaluates to CONS(s0,
〈

s1, . . . , s|s|−1

〉

).

2.5.33 hidel: α listview → α seq

Let lv be a listview value.

• If lv is NIL, then (hidel lv) evaluates to 〈〉.

• If lv is CONS(x,xs), then hidel lv evaluates to a sequence with length
|xs| + 1 such that s′0 is x and s′i is xsi for all valid indices i into xs.

Chapter 3

SET Signature

3.1 Overview

We begin by describing an abstract notion of a set representation, which extends
the standard mathematical sets. The documentation for the signature that
follows states the behavior of implementations in terms of this abstraction.

Like a mathematical set, a set S is a finite collection of unique elements of
some type and the size of S, denoted by |S|, is the number of elements in that
set. The crucial difference between a set in the mathematical sense and a set
is this library is that a set here is always ordered: for enumeration purposes,
the implementation gives an implicit ordering of the elements. The empty set,
denoted by ∅, is a special set that represents an empty collection, so |∅| = 0.

3.2 Signature Definition

signature SET =

sig

type set

type key

type t = set

structure Seq : SEQUENCE

val empty : set

val singleton : key -> set

val size : set -> int

val equal : set * set -> bool

val iter : (’b * key -> ’b) -> ’b -> set -> ’b

val filter : (key -> bool) -> set -> set

val find : set -> key -> bool

val union : (set * set) -> set

val intersection : (set * set) -> set

val difference : (set * set) -> set

val insert : key -> set -> set

val delete : key -> set -> set

val fromSeq : key Seq.seq -> set

val toSeq : set -> key Seq.seq

val toString : set -> string

end

3.3 Details of Types

3.3.1 set

This is the abstract type representing a set described in Section 3.1.

3.3.2 key

This indicates that each element of a set has to have type key .

3.4 Details of Values

3.4.1 empty: set

empty represents the empty set ∅.

3.4.2 singleton: key → set

For a value x of type key , the expression singleton x evaluates to a set con-
taining exactly x.

3.4.3 size: set → int

If s is a value of type set , then size s evaluates to |s| (i.e., the number of
elements in the set represented by s).

3.4.4 equal: set × set → bool

If s1 and s2 are values of type set , then equal (s1,s2) evaluates to true

if s1 and s2 are identical sets (i.e, they have the exact same set of elements);
otherwise, it evaluates to false.

3.4.5 iter: (β × key → β) → β → set → β

If f is a function, b is a value, and s is a set value, then iter f b s iterates
f with left association on s on an implementation-specified ordering, using b as
the base case. That is to say, iter f b s evaluates to

f(f(. . . f(b, s|s|−1), . . . s1), s0),

where s0, s1, . . . , s|s|−1 are the elements of s listed in the order that the imple-
mentation chooses.

3.4.6 filter: (key → bool) → set → set

If p is a predicate and s is a set value, then filter p s evaluates to the subset
s′ of s such that an element x ∈ s′ if and only if p holds on x.

3.4.7 find: set → key → bool

If s is a set value and k is a key value, then find s k evaluates to a boolean
value indicating whether or not k is a member of s.

3.4.8 union: (set × set) → set

If s and t are set values, union (s,t) evaluates to the set s ∪ t.

3.4.9 intersection: (set × set) → set

If s and t are set values, intersection (s,t) evaluates to the set s ∩ t.

3.4.10 difference: (set × set) → set

If s and t are set values, difference (s,t) evaluates to the set s \ t (i.e. the
set {x ∈ s : s 6∈ t}).

3.4.11 insert: key → set → set

If k is a key value and s is a set, insert k s evaluates to the set s ∪ {k}.

3.4.12 delete: key → set → set

If k is a key value and s is a set, delete k s evaluates to the set s \ {k}.

3.4.13 fromSeq: keySeq. seq → set

If s is a sequence value of type key Seq.seq, then fromSeq s evaluates to the set
containing the elements s0, s1, . . . , s|s|−1. The ordering in the set representation
may differ from the ordering in the sequence representation.

3.4.14 toSeq: set → keySeq. seq

If s is a set value where the elements have type key , then toSeq s evaluates to
the sequence of type key Seq.seq containing all |s| elements of s appearing in
the order of the implementation’s choosing.

20



3.4.15 toString: set → string

If s is a set, toString s evaluates to a string representation of s listing the
elements of s, interleaved with “,”.

Chapter 4

TABLE Signature

4.1 Overview

Abstractly, a table is a set of key-value pairs where the keys are unique. For
this reason, we often think of it as a mapping that associates each key with a
value. Since tables are sets, standard set operations apply on them. We denote
by ∅ an empty table. The size of a table S is the number of keys in S and is
rendered |S| in mathematical notation. Furthermore, a table of size n can be
written as follows

{(k1, v1), (k2, v2), . . . , (kn, vn)},

where k1, . . . , kn are n distinct keys and each key ki maps to vi for i ∈ [n]. For
concreteness, we say that a key k is present in a table T , written as k ∈m T ,
if there exists a value v such that (k, v) ∈ T . The documentation that follows
states the behavior of operations on this abstraction.

4.2 Signature Definition

signature TABLE =

sig

type ’a table

type ’a t = ’a table

structure Key : EQKEY

type key = Key.t

structure Seq : SEQUENCE

type ’a seq = ’a Seq.seq

structure Set : SET where type key = key

type set = Set.set

val empty : unit -> ’a table

val singleton : key * ’a -> ’a table

val size : ’a table -> int

val map : (’a -> ’b) -> ’a table -> ’b table

val mapk : (key * ’a -> ’b) -> ’a table -> ’b table

val tabulate : (key -> ’a) -> set -> ’a table

val domain : ’a table -> set

val range : ’a table -> ’a seq

val reduce : (’a * ’a -> ’a) -> ’a -> ’a table -> ’a

val filter : (key * ’a -> bool) -> ’a table -> ’a table

val iter : (’b * (key * ’a) -> ’b) -> ’b -> ’a table -> ’b

val iterh : (’b * (key * ’a) -> ’b) -> ’b -> ’a table -> (’b table * ’b)

val find : ’a table -> key -> ’a option

val merge : (’a * ’a -> ’a) -> (’a table * ’a table) -> ’a table

val mergeOpt : (’a * ’a -> ’a option) -> (’a table * ’a table) -> ’a table

val extract : ’a table * set -> ’a table

val extractOpt : (’a * ’b -> ’c option) -> ’a table * ’b table -> ’c table

val erase : ’a table * set -> ’a table

val insert : (’a * ’a -> ’a) -> (key * ’a) -> ’a table -> ’a table

val delete : key -> ’a table -> ’a table

val fromSeq : (key*’a) seq -> ’a table

val toSeq : ’a table -> (key*’a) seq

val collect : (key*’a) seq -> ’a seq table

val toString : (’a -> string) -> ’a table -> string

end;

4.3 Details of Types

4.3.1 α table

This is the abstract type representing a table with key type key (see below) and
value type α.

4.3.2 αt

This type is a shorthand for the abstract type α table representing a table.

4.3.3 key

This indicates that the type of keys in a table has to have type key .

4.4 Details of Values

4.4.1 empty: unit → α table

empty represents the empty collection ∅.

4.4.2 singleton: key × α → α table

If k is a value of type key and v is a value of type α, the expression singleton

(k,v) evaluates to the collection {(k, v)}.

4.4.3 size: α table → int

If T is a value of type α table, then size T evaluates to |T | (i.e., the number of
keys in the collection T).

4.4.4 map: (α → β) → α table → β table

If f is a function of type α → β and T is a value of type α table with entries

{(k1, v1), . . . , (kn, vn)},

then map f T evaluates to {(k1, fv1), (k2, fv2), . . . , (kn, fvn)}. That is, it cre-
ates a new collection with the same keys by applying f on each value.

4.4.5 mapk: (key × α → β) → α table → β table

This function generalizes the map function. If f is a function of type key×α → β
and T is a value of type α table with entries {(k1, v1), . . . , (kn, vn)}, then mapk

f T evaluates to {(k1, f(k1, v1)), (k2, f(k2, v2)), . . . , (kn, f(kn, vn))}.

4.4.6 tabulate: (key → α) → set → α table

If f is a function of type key → α and S is a value of type set with elements

{k1, . . . , kn},

then tabulate f S evaluates to {(k1, fk1), (k2, fk2), . . . , (kn, fkn)}.

4.4.7 domain: α table → set

For a table T, the function domain T returns the domain of T as a set.

4.4.8 range: α table → α seq

For a table T, the function range T returns the range of T as a sequence. In
particular it is equivalent to Seq.map (fn (k,v) => v) (toSeq T).

4.4.9 reduce: (α × α → α) → α → α table → α

The function reduce f init T returns the same as Seq.reduce f init (range(T))

4.4.10 filter: (key × α → bool) → α table → α table

If p is a predicate and T is an α table value, then filter p T evaluates to the
collection T ′ of T such that (k, v) ∈ T if and only if p evaluates to true on
(k, v).

4.4.11 iter: (β × (key × α) → β) → β → α table → β

If f is a function, b is a value, and T is a table value, then iter f b s iterates
f with left association on T on an implementation-specified ordering, using b as
the base case. That is, iter f b T evaluates to

f(f(. . . f(b, (k|T |, v|T |)), . . . (k2, v2))), (k1, v1)),

where (k1, v1), (k2, v2), . . . , (k|T |, v|T |) are members of T listed in the order that
the implementation chooses.

4.4.12 iterh: (β×(key×α) → β) → β → α table → (β table×β)

If f is a function, b is a value, and T is a table value, then iterh f b s iterates
f with left association on T on an implementation-specified ordering, using b
as the base case. Unlike iter, iterh also stores intermediate results in a table.
That is, if the implementation orders T as (k1, v1), (k2, v2), . . . , ..., (k|T |, v|T |)
and we let ri denote the result of the partial evaluation up to the i-th pair
(i.e., ri = f(f(. . . f(b, (ki, vi)), . . . (k2, v2))), (k1, v1))), then iterh evaluates to
the pair

({(ki, ri) : i = 1, . . . , |T |}, r|T |),

where r|T | = iter f b T by definition.

4.4.13 find: α table → key → α option

If T is a table value and k is a key value, then find T k evaluates to SOME v

provided that k is present in T and is associated with the value v; otherwise, it
evaluates to NONE.

4.4.14 merge: (α × α → α) → (α table × α table) → α table

merge is a generalization of set union in the following sense. If f is a function
of type α × α → α and S and T are α tables, then merge f (S, T) evaluates
a table with the following properties: (1) it contains all the keys from S and T
and (2) for each key k, its associated value is inherited from either S or T if k is
present in exactly one of them. But if k is present in both tables, i.e., (k, v) ∈ S
and (k, w) ∈ T , then the value is f(v, w).

4.4.15 mergeOpt: (α × α → α option) → (α table × α table) →
α table

mergeOpt further generalizes set union, allowing values to cancel out each other
and eliminate the presence of a key in manner similar to set symmetric difference.
If f is a function of type α×α → α option and S and T are α tables, then merge

f (S, T) evaluates a table with the following properties: (1) it contains all the
keys from S and T and (2) for each key k, its associated value is inherited from
either S or T if k is present in exactly one of them. But if k is present in both
tables, i.e., (k, v) ∈ S and (k, w) ∈ T , then the following outcomes are possible:
in the case that f(v, w) evaluates to NONE, the key k will not be present in
the output table; otherwise, f(v, w) evaluates to SOME r and the key k will be
associated with the value r.

4.4.16 extract: α table × set → α table

extract is a generalization of set intersection in the following sense. If T is an
α table and S is a set, then extract (T,S) evaluates to {(k, v) ∈ T : k ∈m S}.

4.4.17 extractOpt: (α × β → γ option) → α table × β table →
γ table

extractOpt is a further generalization of set intersection. If f is a function
α × β → γ option , T is an α table, and S is a β table, then extractOpt f

(T,S) evaluates to {(k, w) : (k, v) ∈ T, (k, v′) ∈ S, and w = f(v, v′)}.

4.4.18 erase: α table × set → α table

This operation extends set difference. If T is an α table, and S is a set, then
erase (T,S) evaluates to {(k, v) ∈ T : (k, v) ∈ T, k 6∈m S}.

4.4.19 insert: (α × α → α) → (key × α) → α table → α table

For a function f of type α × α → α, a key-value pair (k, v), and a table T,
insert f (k, v) T evaluates to T ∪{(k, v)} provided that k 6∈m T ; otherwise,
if (k, v′) ∈ T , it evaluates to (T \ {(k, v′)}) ∪ {(k, f(v, v′)}} (i.e., it replaces the
value associated with k with the result of applying f on the old value v′ and
the new value v).

4.4.20 delete: key → α table → α table

If k is a value of type key and T is an α table, then delete k T evaluates to
{(k′, v′) ∈ T : k′ 6= k}.

4.4.21 fromSeq: (key × α) seq → α table

If s is a key × α sequence such that

s = 〈(k1, v1), (k2, v2), . . . (kn, vn)〉,

then fromSeq s evaluates to {(k1, v1), (k2, v2), . . . (kn, vn)}.

4.4.22 toSeq: α table → (key × α) seq

If T is an α table representing {(k1, v1), (k2, v2), . . . (kn, vn)}, then toSeq T eval-
uates to 〈(k1, v1), (k2, v2), . . . (kn, vn)〉, where the ordering is determined by the
implementation.

4.4.23 collect: (key × α) seq → α seq table

This function groups values of the same key together as a sequence of values
that respects the original sequence ordering. Specifically, if s is a key × α
sequence representing 〈(k1, v1), (k2, v2), . . . (kn, vn)〉, then collect s evaluates
to {(ℓ1, s1), (ℓ2, s2), . . . , (ℓm, sm)}, where the ℓi’s are unique keys belonging to
{k1, . . . , kn} and for i ∈ [m], si is the sequence of values in s with the key ℓi

(i.e., si = 〈vj : kj = ℓi〉).

